Appendix A Computer Program for the Calculation of Gas Solubility

A computer program has been prepared to calculate gas solubility in a n-paraffin wax based on the correlation of this work. Solubilities are calculated for any one of the gases H_2, CO, CH_4, CO_2, C_2H_4, C_2H_6 or their mixtures at a specified temperature, pressure, and wax-free mole fractions of the gases in the gas phase. The molecular weight of the wax is specified. The calculation gives the solubilities in mols per kg of wax and in mole fractions in the liquid. The equilibrium mole fraction of wax in the vapor is also calculated. The program is in FORTRAN.

The input for the program is illustrated below.

Card no.	Format	Variables
1	2F10.3	T in K and P in atm of the system
2	I1	number of solutes in the system
3	F10.3	molecular weight of the wax
4 and after	I1,f10.3	gas identification and vapor phase composition on wax-free basis

Gas identification is an interger, which represents a gas species studied in this work.

Gas Identification	Gas species
1	hydrogen
2	carbon monoxide
3	methane
4	carbon dioxide
5	ethane

The output lists the calculated liquid phase composition of each solute and th wax mol fraction in the vapor phase.

Program List

```
35005,b2u,t80,ll,pr
mnfftn(t,u)
#eor
      program gas(input,output,tape5=input,
     &tape6=output)
      dimension x(5),f(5)
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
      common con(2,6),xx(6)
      common /was/ ywax
c
c     t: temperature, K
c     p: pressure, atm
c     ngas: number of gas species
c     ncomt: number of total species including wax
c     vcal: calculated gas or liquid volume
c     x: liquid composition
c     y: vapor composition
c     r: gas constant
c     rmw: molecular weight of the wax
c     rn: equivalent carbon number of the wax
c     a,b, alpha: RKS equation parameters
c     tc,pc,omeg: physical properties of species
c     phi: volume fraction of a species
c     chi: adjustable parameter Aij
c     q: van der Waal's volume of a species
c     parm,xjac,xjtj,work,con: working variables
c
      r=82.06
      read(5,2)t,p
    2 format(2f9.3)
      read(5,3)ngas
    3 format(i1)
      ncomt=ngas+1
      read(5,12)rmw
   12 format(f10.3)
      call wax
      write(6,14)t,p
   14 format(1x,'system temperature in k=',f10.2,/,
     &8x,'pressure in atm =',f10.2,//)
      write(6,15)rmw
   15 format(1x,'wax molecular weight=',f10.2,/)
c
c  read in the the gas id, and its vapor-
c  phase composition
```

```
c
      do 20 n1=1,ncomt
      do 20 n2=1,ncomt
   20 chi(n1,n2)=0.0
      write(6,22)
   22 format(/,1x,'species no.',5x,'species',12x,
     &'y on wax-free basis')
      do 39 j1=1,ngas
      read(5,25)id,y(j1)
   25 format(i1,2f10.4)
      x(j1)=y(j1)/10.
      call gprop(id,j1)
   39 continue
      ywax=0.0
      write(6,49)ncomt,ywax
   49 format(4x,i1,12x,'wax   ',12x,f12.4)
      do 50 n1=1,ncomt
   50 chi(ncomt,n1)=chi(n1,ncomt)
c
c  call subroutine to find the liquid composition.
c
      do 100 icont=1,10
      call obj(x,f)
      grad=0.0
      do 60 n1=1,ngas
   60 grad=grad+abs(f(n1)/x(n1)-1.0)
      if(grad-5.0e-5)110,90,90
   90 do 95 n1=1,ngas
   95 x(n1)=f(n1)
  100 continue
  110 write(6,118)
  118 format(//,'species no.',5x,'calc. y',5x,
     &'calc. x',7x,'solubility',/,
     &42x,'(mol per kg wax)',/)
      dumx=0.0
      dumy=0.0
      do 150 j1=1,ngas
      sol=x(j1)/(1.0-x(j1))/rmw*1000.0
      dumx=dumx+con(1,j1)
      dumy=dumy+con(2,j1)
      write(6,120)j1,con(2,j1),x(j1),sol
  120 format(1x,i3,4x,f12.4,3x,e12.3,4x,e12.3)
  150 continue
      dumx=dumx+con(1,ncomt)
      dumy=dumy+con(2,ncomt)
      write(6,130)ncomt,con(2,ncomt),con(1,ncomt)
  130 format(1x,i3,4x,f12.4,3x,e12.3,12x,'-')
```

```
      write(6,140)dumy,dumx
  140 format(/,1x,'total',3x,f12.4,1x,f12.4,/)
      stop
      end

      subroutine obj(x,f)
c
c   this subroutine set the objective function to be
c   minimized by the levenberg-marquardt algorithm
c
      dimension coef(4),fu(2,6),x(5),f(5)
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
      common con(2,6),xx(6)
      common /was/ ywax
      complex z(3)
      con(1,ncomt)=1.0
      con(2,ncomt)=ywax
      do 10 i0=1,ngas
      con(1,i0)=x(i0)
      con(2,i0)=(1.0-ywax)*y(i0)
      con(1,ncomt)=con(1,ncomt)-con(1,i0)
   10 continue
      con(1,ncomt)=abs(con(1,ncomt))
      do 50 j1=1,ncomt
      tr=t/tc(j1)
      call calal(j1,tr)
      a(j1)=0.42747*alpha(j1)*r*r*tc(j1)*tc(j1)/pc(j1)
      b(j1)=0.08664*r*tc(j1)/pc(j1)
   50 continue
      do 90 i1=1,2
      call mix(i1)
      call mixb(i1)
      call mixa(i1)
      rtop=r*t/p
      coef(1)=1.0
      coef(2)=-rtop
      coef(3)=-bm*rtop-bm*bm+am/p
      coef(4)=-bm*am/p
      call zrpoly(coef,3,z,ier)
      vcal(1)=real(z(1))
      vcal(2)=real(z(3))
      do 90 j1=1,ncomt
      call rabrn(i1,j1,dadxi,dbdxi)
      call fugac(fu,vcal(i1),i1,j1,dadxi,dbdxi)
```

```
 90 continue
    do 100 i1=1,ngas
    f(i1)=con(2,i1)/(fu(1,i1)/fu(2,i1))
100 continue
    xwax=1.0
    do 110 n1=1,ngas
110 xwax=xwax-f(n1)
    ywax=fu(1,ncomt)/fu(2,ncomt)*xwax
    return
    end

    subroutine wax
c
c  given the molecular weight of the wax, this
c  subroutine generates the necessary wax
c  information for further calculations
c
    dimension coef(4,4)
    common t,p,y(5),ngas,ncomt,vcal(2),
   &r,rmw,rn,tc(6),pc(6),omeg(6),
   &a(6),b(6),am,bm,alpha(6),
   &phi(6),chi(6,6),q(6)
    common con(2,6),xx(6)
    rn=(rmw-2.0158)/14.0268
    q(ncomt)=13.67*2.0+(rn-2.0)*10.23
    coef(1,1)=0.31391e04
    coef(2,1)=-0.76229e00
    coef(3,1)=0.11165e03
    coef(4,1)=0.16693e-01
    coef(1,2)=0.36175e-01
    coef(2,2)=-0.47775e00
    coef(3,2)=0.58524e00
    coef(4,2)=0.43358e-01
    coef(1,3)=0.80936e01
    coef(2,3)=0.99288e00
    coef(3,3)=0.54555e02
    coef(4,3)=0.80034e01
    coef(1,4)=0.73697e-01
    coef(2,4)=0.14880e00
    coef(3,4)=0.57077e-1
    coef(4,4)=0.44621e-1
    do 10 i=1,4
    c1=coef(1,i)
    c2=coef(2,i)
    c3=coef(3,i)
    c4=coef(4,i)
    dum1=c1/c4
```

```
      dum2=1.0-c2
      term2=dum1-c3**dum2
      term3=exp(-c4*(rn-1.0)*dum2)
      dum=(dum1-term2*term3)**(1.0/dum2)
      goto (3,4,6,8),i
    3 haha=3.03191-0.049901*rn**(2.0/3.0)
      tb=1078.-10.0**haha
      goto 10
    4 tc(ncomt)=tb/dum
      goto 10
    6 pc(ncomt)=dum
      goto 10
    8 omeg(ncomt)=dum
   10 continue
      return
      end

      subroutine gprop(id,j1)
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
      common con(2,6),xx(6)
c
c  this subroutine generates the gas information needed
c  for calcuations
c
      goto (10,20,30,40,50),id
   10 write(6,11)j1,y(j1)
   11 format(4x,i1,12x,'hydrogen ',9x,f12.4)
      tc(j1)=33.2
      pc(j1)=12.8
      omeg(j1)=-0.22
      q(j1)=10.51
      a1=0.2456
      a2=0.2406
      rmw0=127.30
      goto 100
   20 write(6,21)j1,y(j1)
   21 format(4x,i1,12x,'carbon monoxide',3x,f12.4)
      tc(j1)=132.9
      pc(j1)=34.5
      omeg(j1)=0.049
      q(j1)=13.01
      a1=0.1723
      a2=0.1707
      rmw0=140.75
```

```
      goto 100
   30 write(6,31)j1,y(j1)
   31 format(4x,i1,12x,'methane  ',9x,f12.4)
      tc(j1)=190.6
      pc(j1)=45.4
      omeg(j1)=0.008
      q(j1)=17.09
      a1=0.1009
      a2=0.1050
      rmw0=146.06
      goto 100
   40 write(6,41)j1,y(j1)
   41 format(4x,i1,12x,'carbon dioxide ',3x,f12.4)
      tc(j1)=304.2
      pc(j1)=72.8
      omeg(j1)=0.225
      q(j1)=19.68
      a1=0.09710
      a2=0.09240
      rmw0=137.02
      goto 100
   50 write(6,51)j1,y(j1)
   51 format(4x,i1,12x,'ethane   ',9x,f12.4)
      tc(j1)=305.4
      pc(j1)=48.2
      omeg(j1)=0.098
      q(j1)=27.34
      a1=0.04967
      a2=0.05277
      rmw0=188.16
  100 factor=1.0-exp(-rmw/rmw0)
      chi(j1,ncomt)=(a1+(a2-a1)*(t-473.2)/100.0)*factor
      return
      end

      subroutine calal(j1,tr)
c
c     this subroutine calculates equation parameter
c     alpha for each species
c
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
      common con(2,6),xx(6)
      if(tc(j1)-550.)10,10,40
   10 dum=omeg(j1)
```

```
      haha=0.480+1.574*dum-0.176*dum*dum
      haha=1.0+haha*(1.0-sqrt(tr))
      alpha(j1)=haha*haha
      goto 120
   40 if(rmw-268.529)50,50,60
   50 power=0.5
      goto 70
   60 power=0.5685-2.552e-4*rmw
   70 dum=rmw
      ff=0.47135+0.63325e-2*dum-0.78307e-5*dum*dum
     &+0.44238e-8*dum**3-0.97365e-12*dum**4
      alpha(j1)=(1.0+ff*(1.0-tr**power))**(1.0/power)
  120 return
      end

      subroutine mix(i1)
c
c     this subroutine sets the parameters needed
c     for executing the mixing rules.
c
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
      common con(2,6),xx(6)
      dum=0.0
      do 10 n1=1,ncomt
   10 dum=dum+con(i1,n1)*q(n1)
      do 20 n1=1,ncomt
   20 phi(n1)=con(i1,n1)*q(n1)/dum
      dum1=0.0
      do 30 n1=1,ncomt
   30 dum1=dum1+con(i1,n1)
      do 40 n1=1,ncomt
   40 xx(n1)=dum1*q(n1)/dum
      return
      end

      subroutine mixa(i1)
c
c   this subroutine executes the mixing rule
c   for a
c
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
```

```
      common con(2,6),xx(6)
      rlam=alog(2.0)
      sum1=0.0
      do 100 n1=1,ncomt
  100 sum1=sum1+con(i1,n1)*q(n1)
      sum0=0.0
      do 150 n1=1,ncomt
      do 150 n2=1,ncomt
  150 sum0=sum0+chi(n1,n2)*phi(n1)*phi(n2)
      sum1=0.5*sum1*sum0
      sum2=0.0
      do 200 n1=1,ncomt
  200 sum2=sum2+con(i1,n1)*alog(xx(n1))
      ge=(sum1+sum2)*r*t
      sum3=0.0
      do 400 n1=1,ncomt
  400 sum3=sum3+con(i1,n1)*a(n1)/b(n1)
      am=bm*(sum3-ge/rlam)
      return
      end

      subroutine mixb(i1)
c
c  this subroutine executes mixing rule
c  for b
c
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
      common con(2,6),xx(6)
      bm=0.0
      do 100 n1=1,ncomt
      bm=bm+con(i1,n1)*b(n1)
  100 continue
      return
      end

      subroutine rabrn(i1,j1,dadxi,dbdxi)
c
c  this subroutines calcultes the partial
c  derivatives for a and b
c
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
```

```
      common con(2,6),xx(6)
c
c     calculate partial b
c
      dbdxi=b(j1)
c
c     calculate prtial a
c
      rlam=alog(2.0)
      sum0=0.0
      do 100 n1=1,ncomt
  100 sum0=sum0+con(i1,n1)*q(n1)
      sum1=0.0
      do 150 n1=1,ncomt
      do 150 n2=1,ncomt
  150 sum1=sum1+chi(n1,n2)*phi(n1)*phi(n2)
      tum1=0.5*sum0*sum1
      sum2=0.0
      do 200 n1=1,ncomt
  200 sum2=sum2+con(i1,n1)*alog(xx(n1))
      tum2=sum2
      ge=(tum1+tum2)*r*t
      sum3=0.0
      do 400 n1=1,ncomt
  400 sum3=sum3+con(i1,n1)*a(n1)/b(n1)
      term1=b(j1)*(sum3-ge/rlam)
      term21=0.5*sum1*q(j1)
      sum4=0.0
      do 450 n1=1,ncomt
  450 sum4=sum4+chi(n1,j1)*xx(j1)*phi(n1)
      sum5=0.0
      do 470 n1=1,ncomt
      do 470 n2=1,ncomt
  470 sum5=sum5+chi(n1,n2)*phi(n1)*phi(n2)*xx(j1)
      term22=sum0*(sum4-sum5)
      term23=alog(xx(j1))
      sum6=0.0
      do 490 n1=1,ncomt
  490 sum6=sum6+con(i1,n1)*xx(j1)
      term24=1.0-sum6
      dum2=term21+term22+term23+term24
      term2=bm*(a(j1)/b(j1)-r*t/rlam*dum2)
      dadxi=term1+term2
      return
      end

      subroutine fugac(fu,v,i1,j1,dadxi,dbdxi)
```

```
      dimension fu(2,6)
c
c     this subroutine calculates the fugacities
c
      common t,p,y(5),ngas,ncomt,vcal(2),
     &r,rmw,rn,tc(6),pc(6),omeg(6),
     &a(6),b(6),am,bm,alpha(6),
     &phi(6),chi(6,6),q(6)
      common con(2,6),xx(6)
      z=abs(p*v/r/t)
      vdum=abs(v)
      vpb=abs(v+bm)
      vmb=abs(v-bm)
      term1=alog(vdum/vmb)
      term2=(dadxi/bm-am*dbdxi/bm/bm)*alog(vdum/vpb)/r/t
      term3=dbdxi/vmb
      term4=am/bm*dbdxi/vpb/r/t
      philn=term1+term2+term3-term4-alog(z)
      fu(i1,j1)=exp(philn)
      return
      end
#eor
```

----------Sample Input----------

```
573.2     20.00   t(K) and p(atm) of the system; 2f10.3
5                 number of gases; i1
394.77            molecular weight of wax;f10.3
1   0.3000        gas id. and y in mol fraction;i1,2f10.3
2   0.3000
3   0.0500
4   0.3000
5   0.0500
#eof
```

---------- Sample Output ----------

```
system temperature in k=    573.20
     pressure in atm =      20.00

wax molecular weight=      394.77

species 1   gas id= 1
species 2   gas id= 2
species 3   gas id= 3
species 4   gas id= 4
species 5   gas id= 5
```

species	y	calc. x	solubility (mol per kg wax)
1	.3000	.173e-01	.445e-01
2	.3000	.204e-01	.527e-01
3	.0500	.520e-02	.132e-01
4	.3000	.380e-01	.100e+00
5	.0500	.870e-02	.222e-01

wax mole fraction in vapor phase= .165e-02